Perturbation of Stochastic Growth Model

In[964]:=
```
x = 0; ClearAll["Global`*"]; DateList[Date[]] // Most
SetDirectory[NotebookDirectory[]]
```

Out[964]= {2022, 3, 24, 15, 36}

Out[965]= /Users/kennethjudd/Desktop

Stochastic Model Setup

K: One capital stock

c: One output good, used for consumption and investment

θ: Productivity state

θ follows an AR(1) process

z: innovations, mean zero unit variance

ϵ: size of innovation in productivity

In[966]:=
```
kplus = f[k, θ] - c[k, θ, ϵ];
cplus = c[kplus, θplus, ϵ];
θplus = ρ θ + ϵ z;
fp[x_, y_] = D[f[x, y], x];
EulerEq = u'[c[k, θ, ϵ]] - β u'[cplus] × fp[kplus, θplus]
```

Out[970]=
```
u′[c[k, θ, ϵ]] -
  β u′[c[-c[k, θ, ϵ] + f[k, θ], z ϵ + θ ρ, ϵ]] f^(1,0)[-c[k, θ, ϵ] + f[k, θ], z ϵ + θ ρ]
```

Log utility and Cobb-Douglas production function

In[971]:=
```
u[x_] = Log[x];
f[x_, y_] = x + A x^α Exp[y];
```

Steady state conditions for deterministic problem:

```
In[973]:= kss = 1;
          θss = 0;
          ϵss = 0;

In[976]:= ss = {k → kss, θ → θss, ϵ → ϵss, 1. → 1, 0. → 0}

Out[976]= {k → 1, θ → 0, ϵ → 0, 1. → 1, 0. → 0}

In[977]:= kplus - k /. ss

Out[977]= A - c[1, 0, 0]

In[978]:= Solve[% == 0, c[1, 0, 0]]

Out[978]= {{c[1, 0, 0] → A}}

In[979]:= c[1, 0, 0] = A;
```

Euler equation at steady state should be zero.
Need to find a value for A in the production function to make that true -- this is just a normalization, a choice of units.

In[980]:=
```
EulerEq //. ss
```

Out[980]= $\frac{1}{A} - \frac{(1 + A\,\alpha)\,\beta}{A}$

In[981]:=
```
Solve[% == 0, A]
```

Out[981]= $\left\{\left\{A \to \frac{1-\beta}{\alpha\,\beta}\right\}\right\}$

In[982]:=
```
A = A //. %[[1]]
```

Out[982]= $\frac{1-\beta}{\alpha\,\beta}$

Set parameters

In[983]:=
```
α = 1. / 3; β = 95 / 100; ρ = 1 / 2;
```

Display Euler equation

In[984]:=
```
EulerEq
```

Out[984]=

$$\frac{1}{c[k,\theta,\epsilon]}-\frac{19\left(1+\frac{0.0526316\,e^{z\epsilon+\frac{\theta}{2}}}{\left(0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}\right)}{20\,c\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]}$$

In[985]:=
```
EulerEq //. ss
```

Out[985]= 8.88178×10^{-16}

NOTE: For the stochastic case, only the E_z\{EulerEq\} = 0.

which is essentially zero

sol will be list of solutions

In[986]:=
```
sol = {};
```

Perturbations of the deterministic system

We now solve for c[k, θ, 0], which is the deterministic system

k pert

EulerEq must be zero at all k. Therefore, same must be true of its derivatives

In[987]:= `D[EulerEq, k]`

Out[987]=

$$\left(0.0333333\, e^{z\,\epsilon+\frac{\theta}{2}}\left(1+\frac{0.0526316\, e^{\theta}}{k^{0.666667}}-c^{(1,0,0)}[k,\theta,\epsilon]\right)\right)\Big/ \left(\left(0.157895\, e^{\theta}\, k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{1.66667} c\left[0.157895\, e^{\theta}\, k^{0.333333}+k-c[k,\theta,\epsilon],\, z\,\epsilon+\frac{\theta}{2},\, \epsilon\right]\right)-\frac{c^{(1,0,0)}[k,\theta,\epsilon]}{c[k,\theta,\epsilon]^{2}}+$$

$$\left(19\left(1+\frac{0.0526316\, e^{z\,\epsilon+\frac{\theta}{2}}}{\left(0.157895\, e^{\theta}\, k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}\right)\left(1+\frac{0.0526316\, e^{\theta}}{k^{0.666667}}-c^{(1,0,0)}[k,\theta,\epsilon]\right) c^{(1,0,0)}\left[0.157895\, e^{\theta}\, k^{0.333333}+k-c[k,\theta,\epsilon],\, z\,\epsilon+\frac{\theta}{2},\, \epsilon\right]\right)\Big/ \left(20\, c\left[0.157895\, e^{\theta}\, k^{0.333333}+k-c[k,\theta,\epsilon],\, z\,\epsilon+\frac{\theta}{2},\, \epsilon\right]^{2}\right)$$

What a mess BUT we only care about this at the deterministic steady state:

In[988]:=
```
% //. ss
```

Out[988]= $0.211111 \left(1.05263 - c^{(1,0,0)}[1, 0, 0]\right) - 40.1111\, c^{(1,0,0)}[1, 0, 0] + 40.1111 \left(1.05263 - c^{(1,0,0)}[1, 0, 0]\right) c^{(1,0,0)}[1, 0, 0]$

In[989]:=
```
% // Simplify
```

Out[989]= $0.222222 + 1.9\, c^{(1,0,0)}[1, 0, 0] - 40.1111\, c^{(1,0,0)}[1, 0, 0]^2$

A quadratic equation. Let's find the solutions

In[990]:= $\text{Solve}\left[\% == 0,\ c^{(1,0,0)}[1, 0, 0]\right]$

Out[990]= $\left\{\left\{c^{(1,0,0)}[1, 0, 0] \to -0.0544254\right\}, \left\{c^{(1,0,0)}[1, 0, 0] \to 0.101794\right\}\right\}$

Two solutions BUT we know that consumption is increasing in k implying

In[991]:=
```
sol = Union[sol, %[[2]]]
```

Out[991]= $\left\{c^{(1,0,0)}[1, 0, 0] \to 0.101794\right\}$

θ pert

θ is an exogenous state. We now want to see how θ affects consumption. EulerEq must be zero for all θ. Therefore, so are all of its derivatives

In[992]:=
```
D[EulerEq, θ]
```

Out[992]=

$$-\frac{19\left(\frac{0.0263158\, e^{z\epsilon+\frac{\theta}{2}}}{\left(0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}-\frac{0.0350877\, e^{z\epsilon+\frac{\theta}{2}}\left(0.157895\, e^{\theta} k^{0.333333}-c^{(0,1,0)}[k,\theta,\epsilon]\right)}{\left(0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{1.66667}}\right)}{20\, c\left[0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon],\ z\epsilon+\frac{\theta}{2},\ \epsilon\right]}-\frac{c^{(0,1,0)}[k,\theta,\epsilon]}{c[k,\theta,\epsilon]^{2}}+\left(19\left(1+\frac{0.0526316\, e^{z\epsilon+\frac{\theta}{2}}}{\left(0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}\right)\left(\frac{1}{2}\, c^{(0,1,0)}\left[0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon],\ z\epsilon+\frac{\theta}{2},\ \epsilon\right]+\left(0.157895\, e^{\theta} k^{0.333333}-c^{(0,1,0)}[k,\theta,\epsilon]\right) c^{(1,0,0)}\left[0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon],\ z\epsilon+\frac{\theta}{2},\ \epsilon\right]\right)\right)\Big/\left(20\, c\left[0.157895\, e^{\theta} k^{0.333333}+k-c[k,\theta,\epsilon],\ z\epsilon+\frac{\theta}{2},\ \epsilon\right]^{2}\right)$$

Mess BUT we only care about steady state

In[993]:= `% //. ss // Simplify`

Out[993]= $-0.125 + c^{(0,1,0)}[1, 0, 0] \left(-20.2667 - 40.1111\, c^{(1,0,0)}[1, 0, 0]\right) + 6.33333\, c^{(1,0,0)}[1, 0, 0]$

We have already computed $c^{(1,0,0)}[1, 0, 0]$, so we use that solution

In[994]:= `% //. sol // Simplify`

Out[994]= $0.519694 - 24.3497\, c^{(0,1,0)}[1, 0, 0]$

Now solve for the unknown, the gradient of c w.r.t. θ at the steady state

In[995]:= `Solve[% == 0, c^(0,1,0)[1, 0, 0]]`

Out[995]= $\{\{c^{(0,1,0)}[1, 0, 0] \to 0.0213429\}\}$

In[996]:= `sol = Union[sol, %[[1]]]`

Out[996]= $\{c^{(0,1,0)}[1, 0, 0] \to 0.0213429,\ c^{(1,0,0)}[1, 0, 0] \to 0.101794\}$

k, θ pert - degree 2

We now analyze the second-order derivatives in (k, θ)

(k,k)

In[997]:=
```
D[EulerEq, k, k] //. ss // Simplify
% //. sol // Simplify
```

Out[997]= $-0.397271 + 1066.96\, c^{(1,0,0)}[1, 0, 0]^3 - 508.074\, c^{(1,0,0)}[1, 0, 0]^4 + c^{(1,0,0)}[1, 0, 0]\left(-3.62963 - 124.556\, c^{(2,0,0)}[1, 0, 0]\right) + 4.12222\, c^{(2,0,0)}[1, 0, 0] + c^{(1,0,0)}[1, 0, 0]^2\left(-49.6111 + 40.1111\, c^{(2,0,0)}[1, 0, 0]\right)$

Out[998]= $-0.209957 - 8.14113\, c^{(2,0,0)}[1, 0, 0]$

In[999]:= $\text{Solve}\left[\% == 0,\ c^{(2,0,0)}[1, 0, 0]\right]$

Out[999]= $\left\{\left\{c^{(2,0,0)}[1, 0, 0] \to -0.0257897\right\}\right\}$

In[1000]:=
```
sol = Union[sol, %[[1]]] // Simplify;
```

(k,θ)

In[1001]:=
```
D[EulerEq, k, θ] //. ss // Simplify
% //. sol // Simplify
```

Out[1001]=
$$0.0637427 - 85.025\, c^{(1,0,0)}[1, 0, 0]^2 + 80.2222\, c^{(1,0,0)}[1, 0, 0]^3 - 19.2111\, c^{(1,1,0)}[1, 0, 0] + c^{(1,0,0)}[1, 0, 0] \left(2.67222 - 60.1667\, c^{(1,1,0)}[1, 0, 0] - 6.33333\, c^{(2,0,0)}[1, 0, 0]\right) + 6.66667\, c^{(2,0,0)}[1, 0, 0] + c^{(0,1,0)}[1, 0, 0] \left(-0.333333 + 786.178\, c^{(1,0,0)}[1, 0, 0]^2 - 508.074\, c^{(1,0,0)}[1, 0, 0]^3 - 42.2222\, c^{(2,0,0)}[1, 0, 0] + c^{(1,0,0)}[1, 0, 0] \left(243.798 + 40.1111\, c^{(2,0,0)}[1, 0, 0]\right)\right)$$

Out[1002]=
$$0.0900211 - 25.3357\, c^{(1,1,0)}[1, 0, 0]$$

In[1003]:=
$$\text{Solve}\left[\% == 0,\ c^{(1,1,0)}[1, 0, 0]\right]$$

Out[1003]=
$$\left\{\left\{c^{(1,1,0)}[1, 0, 0] \to 0.00355313\right\}\right\}$$

In[1004]:=
```
sol = Union[sol, %[[1]]];
```

(θ, θ)

In[1005]:=
```
D[EulerEq, θ, θ] //. ss // Simplify
% //. sol // Simplify
```

Out[1005]=
$$-0.0212719 + c^{(0,2,0)}[1,0,0]\left(-30.2944 - 40.1111\, c^{(1,0,0)}[1,0,0]\right) + 6.58333\, c^{(1,0,0)}[1,0,0] - 12.6667\, c^{(1,0,0)}[1,0,0]^2 + 6.33333\, c^{(1,1,0)}[1,0,0] + c^{(0,1,0)}[1,0,0]\left(0.691667 - 81.3833\, c^{(1,0,0)}[1,0,0] + 160.444\, c^{(1,0,0)}[1,0,0]^2 - 40.1111\, c^{(1,1,0)}[1,0,0] - 12.6667\, c^{(2,0,0)}[1,0,0]\right) + 1.\, c^{(2,0,0)}[1,0,0] + c^{(0,1,0)}[1,0,0]^2 \left(382.041 + 505.4\, c^{(1,0,0)}[1,0,0] - 508.074\, c^{(1,0,0)}[1,0,0]^2 + 40.1111\, c^{(2,0,0)}[1,0,0]\right)$$

Out[1006]=
$$0.586289 - 34.3775\, c^{(0,2,0)}[1,0,0]$$

In[1007]:=
```
Solve[% == 0, c^(0,2,0)[1, 0, 0]]
```

Out[1007]=
$$\left\{\left\{c^{(0,2,0)}[1,0,0] \to 0.0170544\right\}\right\}$$

In[1008]:=
```
sol = Union[sol, %[[1]]]
```

Out[1008]=
$$\left\{c^{(0,1,0)}[1,0,0] \to 0.0213429,\ c^{(0,2,0)}[1,0,0] \to 0.0170544,\ c^{(1,0,0)}[1,0,0] \to 0.101794,\ c^{(1,1,0)}[1,0,0] \to 0.00355313,\ c^{(2,0,0)}[1,0,0] \to -0.0257897\right\}$$

k,θ pert - degree 3

Next we doe the third-order derivatives

```
In[1009]:= D[EulerEq, k, k, k] //. ss // Simplify;
% //. sol // Simplify;

In[1011]:= Solve[% == 0, c^(3,0,0)[1, 0, 0]];

In[1012]:= sol = Union[sol, %[[1]]];

In[1013]:= D[EulerEq, k, k, θ] //. ss // Simplify;
% //. sol // Simplify;

In[1015]:= Solve[% == 0, c^(2,1,0)[1, 0, 0]];

In[1016]:= sol = Union[sol, %[[1]]];
```

In[1017]:=
```
D[EulerEq, k, θ, θ] //. ss // Simplify;
% //. sol // Simplify;
```

In[1019]:=
```
Solve[% == 0, c^(1,2,0)[1, 0, 0]];
```

In[1020]:=
```
sol = Union[sol, %[[1]]];
```

In[1021]:=
```
D[EulerEq, θ, θ, θ] //. ss // Simplify;
% //. sol // Simplify;
```

In[1023]:=
```
Solve[% == 0, c^(0,3,0)[1, 0, 0]];
```

In[1024]:=
```
sol = Union[sol, %[[1]]];
```

In[1025]:=
```
sol
```

Out[1025]=
$$\{c^{(0,1,0)}[1,0,0] \to 0.0213429,\ c^{(0,2,0)}[1,0,0] \to 0.0170544,\ c^{(0,3,0)}[1,0,0] \to 0.015388,\ c^{(1,0,0)}[1,0,0] \to 0.101794,\ c^{(1,1,0)}[1,0,0] \to 0.00355313,\ c^{(1,2,0)}[1,0,0] \to 0.00259483,\ c^{(2,0,0)}[1,0,0] \to -0.0257897,\ c^{(2,1,0)}[1,0,0] \to -0.00244427,\ c^{(3,0,0)}[1,0,0] \to 0.0379959\}$$

Perturbation with respect to the stochastic component

We now examine how c[k, θ, ϵ] depends on ϵ for (k, θ, ϵ) close to (kss, θss, ϵss) = (1, 0, 0)

REMEMBER: For the stochastic case, only E_z{EulerEq} = 0. We take derivatives w.r.t. ϵ first and then impose the expectations operator.

ϵ pert

First - order

In[1026]:= D[EulerEq, ϵ]

Out[1026]=

$$-\frac{c^{(0,0,1)}[k,\theta,\epsilon]}{c[k,\theta,\epsilon]^2}-\frac{19\left(\frac{0.0526316\,e^{z\epsilon+\frac{\theta}{2}}\,z}{\left(0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}+\frac{0.0350877\,e^{z\epsilon+\frac{\theta}{2}}\,c^{(0,0,1)}[k,\theta,\epsilon]}{\left(0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{1.66667}}\right)}{20\,c\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]}+$$

$$\left(19\left(1+\frac{0.0526316\,e^{z\epsilon+\frac{\theta}{2}}}{\left(0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon]\right)^{0.666667}}\right)\right.$$

$$\left(c^{(0,0,1)}\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]+\right.$$

$$z\,c^{(0,1,0)}\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]-$$

$$\left.\left.c^{(0,0,1)}[k,\theta,\epsilon]\,c^{(1,0,0)}\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]\right)\right)\Big/$$

$$\left(20\,c\left[0.157895\,e^{\theta}\,k^{0.333333}+k-c[k,\theta,\epsilon],\,z\epsilon+\frac{\theta}{2},\,\epsilon\right]^2\right)$$

Messy BUT

In[1027]:=
```
% //. ss // Simplify
```

Out[1027]= $z\left(-0.316667+40.1111\, c^{(0,1,0)}[1,0,0]\right)+ c^{(0,0,1)}[1,0,0]\left(-0.211111-40.1111\, c^{(1,0,0)}[1,0,0]\right)$

Impose previous solutions

In[1028]:=
```
% //. sol // Simplify
```

Out[1028]= $0.539421\, z-4.29417\, c^{(0,0,1)}[1,0,0]$

We now take the expectation. This done by setting z = 0

In[1029]:=
```
% //. z → 0 // Chop
```

Out[1029]= $-4.29417\, c^{(0,0,1)}[1,0,0]$

```
Solve[% == 0, c^(0,0,1)[1, 0, 0]] // Chop
```

Out[1030]= $\{\{c^{(0,0,1)}[1,0,0]\to 0.\}\}$

In[1032]:=
```
sol = Union[sol, %[[1]]];
```

For every derivative w.r.t. ϵ once, we get zero. Let's add that fact to our solution set

In[1033]:=
```
sol = Union[sol, {c^(i_,j_,1)[1, 0, 0] → 0}];
```

In[1034]:=
```
sol
```

Out[1034]=
$\{c^{(0,0,1)}[1, 0, 0] \to 0,\ c^{(0,1,0)}[1, 0, 0] \to 0.0213429,\ c^{(0,2,0)}[1, 0, 0] \to 0.0170544,$
$c^{(0,3,0)}[1, 0, 0] \to 0.015388,\ c^{(1,0,0)}[1, 0, 0] \to 0.101794,\ c^{(1,1,0)}[1, 0, 0] \to 0.00355313,$
$c^{(1,2,0)}[1, 0, 0] \to 0.00259483,\ c^{(2,0,0)}[1, 0, 0] \to -0.0257897,$
$c^{(2,1,0)}[1, 0, 0] \to -0.00244427,\ c^{(3,0,0)}[1, 0, 0] \to 0.0379959,\ c^{(i\_,j\_,1)}[1, 0, 0] \to 0\}$

ϵ pert - second order

In[1035]:=
```
D[EulerEq, ϵ, ϵ] //. ss // Simplify
```

Out[1035]=
$$80.2222\, z\, c^{(0,1,1)}[1,0,0] + z^2 \left(-0.316667 + 4.01111\, c^{(0,1,0)}[1,0,0] - 508.074\, c^{(0,1,0)}[1,0,0]^2 + 40.1111\, c^{(0,2,0)}[1,0,0]\right) + c^{(0,0,2)}[1,0,0]\left(-0.211111 - 40.1111\, c^{(1,0,0)}[1,0,0]\right) - 80.2222\, c^{(0,0,1)}[1,0,0]\, c^{(1,0,1)}[1,0,0] + z\, c^{(0,0,1)}[1,0,0] \left(3.58889 - 4.01111\, c^{(1,0,0)}[1,0,0] + c^{(0,1,0)}[1,0,0]\left(-1013.47 + 1016.15\, c^{(1,0,0)}[1,0,0]\right) - 80.2222\, c^{(1,1,0)}[1,0,0]\right) + c^{(0,0,1)}[1,0,0]^2 \left(2.32222 + 1013.47\, c^{(1,0,0)}[1,0,0] - 508.074\, c^{(1,0,0)}[1,0,0]^2 + 40.1111\, c^{(2,0,0)}[1,0,0]\right)$$

In[1036]:=
```
% //. sol // Simplify
```

Out[1036]=
$$0. + 0.221577\, z^2 - 4.29417\, c^{(0,0,2)}[1,0,0]$$

In[1037]:=
```
% // Expand
```

Out[1037]=
$$0. + 0.221577\, z^2 - 4.29417\, c^{(0,0,2)}[1,0,0]$$

Take expectation of each term. We assume z had unit variance. Therefore,

In[1038]:=
```
% //. z^2 → 1
```
Out[1038]=
$$0.221577 - 4.29417\, c^{(0,0,2)}[1, 0, 0]$$

In[1039]:=
```
% //. z → 0
```
Out[1039]=
$$0.221577 - 4.29417\, c^{(0,0,2)}[1, 0, 0]$$

In[1040]:=
```
Solve[% == 0, c^(0,0,2)[1, 0, 0]]
```
Out[1040]=
$$\{\{c^{(0,0,2)}[1, 0, 0] \to 0.0515994\}\}$$

In[1041]:=
```
sol = Union[sol, %[[1]]];
```

In[1042]:=
```
sol
```
Out[1042]=
$$\{c^{(0,0,1)}[1, 0, 0] \to 0,\ c^{(0,0,2)}[1, 0, 0] \to 0.0515994,\ c^{(0,1,0)}[1, 0, 0] \to 0.0213429,$$
$$c^{(0,2,0)}[1, 0, 0] \to 0.0170544,\ c^{(0,3,0)}[1, 0, 0] \to 0.015388,$$
$$c^{(1,0,0)}[1, 0, 0] \to 0.101794,\ c^{(1,1,0)}[1, 0, 0] \to 0.00355313,$$
$$c^{(1,2,0)}[1, 0, 0] \to 0.00259483,\ c^{(2,0,0)}[1, 0, 0] \to -0.0257897,$$
$$c^{(2,1,0)}[1, 0, 0] \to -0.00244427,\ c^{(3,0,0)}[1, 0, 0] \to 0.0379959,\ c^{(i\_,j\_,1)}[1, 0, 0] \to 0\}$$

ϵ pert - third order

In[1043]:=
```
D[EulerEq, ϵ, ϵ, ϵ] //. ss // Simplify;
% //. sol
```

Out[1044]=

$$0. + 9653.41\,(0. + 0.0213429\,z)^3 - 0.0326796\,z + 0.300562\,z^3 - 76.2111\,(0. + 0.0213429\,z)^2\,(0. + 1.\,z) + 1524.22\,(0. - 0.0213429\,z)\,\left(0.0463469 + 0.0170544\,z^2\right) + 6.01667\,(0. + 1.\,z)\,\left(0.0463469 + 0.0170544\,z^2\right) + 6.01667\,(0. + 0.0213429\,z)\,\left(0.0343996 + 1.\,z^2\right) - 4.29417\,c^{(0,0,3)}[1, 0, 0] + z\,\left(-0.0220618 + 120.333\,c^{(0,1,2)}[1, 0, 0]\right)$$

We need to expand this polynomial to make sure that we have all the powers of z (moments of z) isolated

In[1045]:=
```
% // Expand
```

Out[1045]=

$$0. - 1.2792\,z + 0.035918\,z^3 - 4.29417\,c^{(0,0,3)}[1, 0, 0] + 120.333\,z\,c^{(0,1,2)}[1, 0, 0]$$

We now assign the moments starting with the high ones. We do not want to start with setting z = 0:

In[1046]:=
```
% //. z^3 → λ
```

Out[1046]= $0. - 1.2792\, z + 0.035918\, \lambda - 4.29417\, c^{(0,0,3)}[1, 0, 0] + 120.333\, z\, c^{(0,1,2)}[1, 0, 0]$

In[1047]:=
```
% //. z^2 → 1
```

Out[1047]= $0. - 1.2792\, z + 0.035918\, \lambda - 4.29417\, c^{(0,0,3)}[1, 0, 0] + 120.333\, z\, c^{(0,1,2)}[1, 0, 0]$

In[1048]:=
```
% //. z → 0
```

Out[1048]= $0. + 0.035918\, \lambda - 4.29417\, c^{(0,0,3)}[1, 0, 0]$

In[1049]:=
```
Solve[% == 0, c^(0,0,3)[1, 0, 0]]
```

Out[1049]= $\{\{c^{(0,0,3)}[1, 0, 0] \to -0.232874\, (0. - 0.035918\, \lambda)\}\}$

In[1050]:=
```
sol = Union[sol, %[[1]]];
```

We could go on forever. We stop here

Some cross-partials

(ϵ,ϵ,k) pert

In[1051]:=
```
D[EulerEq, ϵ, ϵ, k] //. ss // Simplify
% //. sol // Simplify
```

Out[1051]=
```
-1.03704 c^(0,0,2)[1, 0, 0] - 22.9407 c^(0,0,2)[1, 0, 0] c^(1,0,0)[1, 0, 0] +
 1040.21 c^(0,0,2)[1, 0, 0] c^(1,0,0)[1, 0, 0]^2 - 508.074 c^(0,0,2)[1, 0, 0] c^(1,0,0)[1, 0, 0]^3 -
 80.2222 c^(1,0,1)[1, 0, 0]^2 + 1.9 c^(1,0,2)[1, 0, 0] - 80.2222 c^(1,0,0)[1, 0, 0] c^(1,0,2)[1, 0, 0] +
 z^2 (0.222222 + 1.9 c^(1,0,0)[1, 0, 0] - 2.00556 c^(1,0,0)[1, 0, 0]^2 +
    c^(0,1,0)[1, 0, 0]^2 (17.8272 + 10144.5 c^(1,0,0)[1, 0, 0] - 9653.41 c^(1,0,0)[1, 0, 0]^2) +
    c^(0,2,0)[1, 0, 0] (-1.40741 - 533.478 c^(1,0,0)[1, 0, 0] + 508.074 c^(1,0,0)[1, 0, 0]^2) +
    4.22222 c^(1,1,0)[1, 0, 0] - 4.01111 c^(1,0,0)[1, 0, 0] c^(1,1,0)[1, 0, 0] +
    c^(0,1,0)[1, 0, 0] (-2.81481 + 50.8074 c^(1,0,0)[1, 0, 0]^2 - 1069.63 c^(1,1,0)[1, 0, 0] +
       c^(1,0,0)[1, 0, 0] (-50.8074 + 1016.15 c^(1,1,0)[1, 0, 0])) +
    42.2222 c^(1,2,0)[1, 0, 0] - 40.1111 c^(1,0,0)[1, 0, 0] c^(1,2,0)[1, 0, 0]) -
 42.2222 c^(0,0,2)[1, 0, 0] c^(2,0,0)[1, 0, 0] + 40.1111 c^(0,0,2)[1, 0, 0]
  c^(1,0,0)[1, 0, 0] c^(2,0,0)[1, 0, 0] + c^(0,0,1)[1, 0, 0]
  (c^(1,0,1)[1, 0, 0] (-45.8815 + 4160.86 c^(1,0,0)[1, 0, 0] - 3048.44 c^(1,0,0)[1, 0, 0]^2 +
       80.2222 c^(2,0,0)[1, 0, 0]) + (-84.4444 + 80.2222 c^(1,0,0)[1, 0, 0]) c^(2,0,1)[1, 0, 0]) +
 z (-2.81481 c^(0,1,1)[1, 0, 0] - 1066.96 c^(0,1,1)[1, 0, 0] c^(1,0,0)[1, 0, 0] +
    1016.15 c^(0,1,1)[1, 0, 0] c^(1,0,0)[1, 0, 0]^2 + 3.8 c^(1,0,1)[1, 0, 0] -
```

In[1053]:= `% // Expand`

Out[1053]= $0.405267 - 0.152533\, z^2 - 6.26612\, c^{(1,0,2)}[1, 0, 0]$

In[1054]:= `% //. z^3 → λ;`

In[1055]:= `% //. z^2 → 1;`

In[1056]:= `% //. z → 0`

Out[1056]= $0.252734 - 6.26612\, c^{(1,0,2)}[1, 0, 0]$

In[1057]:= `Solve[% == 0, c^(1,0,2)[1, 0, 0]]`

Out[1057]= $\{\{c^{(1,0,2)}[1, 0, 0] \to 0.0403334\}\}$

In[1058]:= `sol = Union[sol, %[[1]]]`

Out[1058]= $\{c^{(0,0,1)}[1, 0, 0] \to 0,\ c^{(0,0,2)}[1, 0, 0] \to 0.0515994,$
$c^{(0,0,3)}[1, 0, 0] \to -0.232874\,(0. - 0.035918\,\lambda),\ c^{(0,1,0)}[1, 0, 0] \to 0.0213429,$
$c^{(0,2,0)}[1, 0, 0] \to 0.0170544,\ c^{(0,3,0)}[1, 0, 0] \to 0.015388,\ c^{(1,0,0)}[1, 0, 0] \to 0.101794,$
$c^{(1,0,2)}[1, 0, 0] \to 0.0403334,\ c^{(1,1,0)}[1, 0, 0] \to 0.00355313,$
$c^{(1,2,0)}[1, 0, 0] \to 0.00259483,\ c^{(2,0,0)}[1, 0, 0] \to -0.0257897,$
$c^{(2,1,0)}[1, 0, 0] \to -0.00244427,\ c^{(3,0,0)}[1, 0, 0] \to 0.0379959,\ c^{(i\_,j\_,1)}[1, 0, 0] \to 0\}$

(ϵ,ϵ,θ) pert

In[1059]:=
```
D[EulerEq, ϵ, ϵ, θ] //. ss // Simplify;
% //. sol // Simplify
```

Out[1060]= $0.242199 - 0.00394657\, z^2 - 24.3497\, c^{(0,1,2)}[1, 0, 0]$

In[1061]:=
```
% // Expand
```

Out[1061]= $0.242199 - 0.00394657\, z^2 - 24.3497\, c^{(0,1,2)}[1, 0, 0]$

In[1062]:=
```
% //. z^3 → λ;
```

In[1063]:=
```
% //. z^2 → 1;
```

In[1064]:=
```
% //. z → 0
```

Out[1064]= $0.238252 - 24.3497\, c^{(0,1,2)}[1, 0, 0]$

In[1065]:=
```
Solve[% == 0, c^(0,1,2)[1, 0, 0]]
```

Out[1065]= $\{\{c^{(0,1,2)}[1, 0, 0] \to 0.00978459\}\}$

In[1066]:=
```
sol = Union[sol, %[[1]]];
```

Approximation

We have now computed all third-order derivatives. We can construct the third-degree Taylor series around the steady state (1, 0, 0) using the solutions

In[1067]:=
```
Series[c[1 + δ (k - 1) , δ θ, δ ϵ], {δ, 0, 3}] //. sol
```

Out[1067]=
$$0.157895 + (0.101794\,(-1+k) + 0.0213429\,\theta)\,\delta + \left(-0.0128949\,(-1+k)^2 + 0.0257997\,\epsilon^2 + 0.00355313\,(-1+k)\,\theta + 0.00852722\,\theta^2\right)\delta^2 + \left(0.00633266\,(-1+k)^3 + 0.0201667\,(-1+k)\,\epsilon^2 - 0.00122213\,(-1+k)^2\,\theta + 0.0048923\,\epsilon^2\,\theta + 0.00129741\,(-1+k)\,\theta^2 + 0.00256466\,\theta^3 - 0.0388123\,\epsilon^3\,(0. - 0.035918\,\lambda)\right)\delta^3 + O[\delta]^4$$

In[1068]:=
```
% // Normal;
% //. δ → 1
```

Out[1069]=
$$0.157895 + 0.101794\,(-1+k) - 0.0128949\,(-1+k)^2 + 0.00633266\,(-1+k)^3 + 0.0257997\,\epsilon^2 + 0.0201667\,(-1+k)\,\epsilon^2 + 0.0213429\,\theta + 0.00355313\,(-1+k)\,\theta - 0.00122213\,(-1+k)^2\,\theta + 0.0048923\,\epsilon^2\,\theta + 0.00852722\,\theta^2 + 0.00129741\,(-1+k)\,\theta^2 + 0.00256466\,\theta^3 - 0.0388123\,\epsilon^3\,(0. - 0.035918\,\lambda)$$

In[1070]:=
```
c[k_, θ_, ϵ_] = %
```

Out[1070]=
$$0.157895 + 0.101794\,(-1+k) - 0.0128949\,(-1+k)^2 + 0.00633266\,(-1+k)^3 + 0.0257997\,\epsilon^2 + 0.0201667\,(-1+k)\,\epsilon^2 + 0.0213429\,\theta + 0.00355313\,(-1+k)\,\theta - 0.00122213\,(-1+k)^2\,\theta + 0.0048923\,\epsilon^2\,\theta + 0.00852722\,\theta^2 + 0.00129741\,(-1+k)\,\theta^2 + 0.00256466\,\theta^3 - 0.0388123\,\epsilon^3\,(0. - 0.035918\,\lambda)$$